query: #defaultView:BubbleChart
  #Based on https://w.wiki/KG$
  SELECT ?word ?count 
  WITH
  {  SELECT ?formatterstring 
     {  wd:Q87572880 wdt:P1181 ?formatterstring_raw .
        BIND(REPLACE(str(?formatterstring_raw), "1", " ") as ?formatterstring)
     } 
  } as %a0
  WITH
  {  SELECT ?title  (CONCAT( REPLACE(str(?title),"[\\.:,\\[\\]()$]",""), ?formatterstring) as ?input) 
     { INCLUDE %a0 
        ?x wdt:P31 wd:Q13442814 ;
           wdt:P1476 ?title.
        ?x wdt:P50 ?author .
        { ?author ?property wd:Q213439. }
        UNION
        {
          ?author ?property2 _:b9.
          _:b9 wdt:P361 wd:Q213439.
        }
       FILTER(lang(?title)="en") 
     }
   LIMIT 5000
  } as %a
  WITH
  {    SELECT ?re ?nv { ?n wdt:P5176 []; wdt:P1181 ?nv . FILTER( ?nv > 1) FILTER( ?nv < 51) 
       BIND( CONCAT( "^([^ ]+ ){", str( ?nv - 1 ),"}([^ ]+) .*") as ?re) }
  } as %nb                                                            
  WITH
  { SELECT DISTINCT ?word (COUNT(DISTINCT ?title) as ?count) 
                    { 
                      INCLUDE %a  INCLUDE %nb  
                              FILTER (strlen(?word) > 4 )  BIND( LCASE(REPLACE(?input, ?re, "$2")) as ?word) }
    GROUP BY ?word
    HAVING(?count > 5)
    LIMIT 42
  } as %b 
  WHERE
  {
      INCLUDE %b
              
  #    FILTER ( ?word != "from" && ?word != "with" && ?word != "after"
  #           )               
  }
  ORDER BY DESC(?count)
  